
Iteraptor
Iterating Nested Terms Like I’m Five
The tiny elixir library to iterate/map/reduce/filter deeply nested structures in Elixir.
This package is a sibling of Ruby Iteraptor gem.
TL;DR
	source code;
	documentation.

Intro
Iterating both maps and lists in Elixir is charming. One might chain iterators,
map, reduce, filter, select, reject, zip... Everybody having at least eight
hours of experience with Elixir has definitely seen (and even maybe written)
something like this:
~w|aleksei saverio|
|> Enum.map(& String.capitalize/1)
|> Enum.each(fn capitalized_name ->
     IO.puts "Hello, #{capitalized_name}!"
   end)
That is really handy. The things gets cumbersome when it comes to deeply nested
structures, like a map having nested keywords, lists etc. The good example of
that would be any configuration file, having nested subsections.
While Elixir provides helpers to update elements deeply inside such a term:
	Kernel.get_in/2
	Kernel.put_in/{2,3}
	Kernel.update_in/{2,3}
	Kernel.get_and_update_in/{2,3}

all the above would work if and only all the parent levels in the structure exist.
The exception would be get_in/2 which is happily returning nil being asked
for whatever inexisting.
The amount of questions on Stack Overflow asking “how would I modify a nested
structure” forced me to finally create this library. The implementation in Elixir
looks a bit more convoluted since everything is immutable and one cannot just
traverse a structure down to leaves, modifying whatever needed in-place.
The iteration-wide accumulator is required.
That is probably the only example I met in my life where mutability makes things
easier. As a bonus the implementation of bury/4 to store the value deeply inside
a structure, creating the intermediate keys as necessary, was introduced.
It behaves as a proposed but rejected in ruby core
Hash#bury.

So, welcome the library that makes the iteration of any nested map/keyword/list
combination almost as easy as the natural Elixir map and each.
• Iteraptor
Features
	Iteraptor.each/3
to iterate a deeply nested map/list/keyword;
	Iteraptor.map/3
to map a deeply nested map/list/keyword;
	Iteraptor.reduce/4
to reduce a deeply nested map/list/keyword;
	Iteraptor.map_reduce/4
to map and reduce a deeply nested map/list/keyword;
	Iteraptor.filter/3
to filter a deeply nested map/list/keyword;
	Iteraptor.to_flatmap/2
to flatten a deeply nested map/list/keyword into
flatten map with concatenated keys;
	Iteraptor.from_flatmap/3
to “unveil”/“unflatten” the previously flattened map into nested structure;
	use Iteraptor.Iteraptable
to automagically implement Enumerable and Collectable protocols, as well as
Access behaviour on the structure.

Words are cheap, show me the code
Iterating, Mapping, Reducing
# each
iex> %{a: %{b: %{c: 42}}} |> Iteraptor.each(&IO.inspect/1, yield: :all)
# {[:a], %{b: %{c: 42}}}
# {[:a, :b], %{c: 42}}
# {[:a, :b, :c], 42}
%{a: %{b: %{c: 42}}}

# map
iex> %{a: %{b: %{c: 42}}} |> Iteraptor.map(fn {k, _} -> Enum.join(k) end)
%{a: %{b: %{c: "abc"}}}

iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.map(fn
...>      {[_], _} = self -> self
...>      {[_, _], _} -> "YAY"
...>    end, yield: :all)
%{a: %{b: "YAY"}}

# reduce
iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.reduce([], fn {k, _}, acc ->
...>      [Enum.join(k, "_") | acc]
...>    end, yield: :all)
...> |> :lists.reverse()
["a", "a_b", "a_b_c"]

# map-reduce
iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.map_reduce([], fn
...>      {k, %{} = v}, acc -> {​{k, v}, [Enum.join(k, ".") | acc]}
...>      {k, v}, acc -> {​{k, v * 2}, [Enum.join(k, ".") <> "=" | acc]}
...>    end, yield: :all)
{​%{a: %{b: %{c: 42}}}, ["a.b.c=", "a.b", "a"]}

# filter
iex> %{a: %{b: 42, e: %{f: 3.14, c: 42}, d: %{c: 42}}, c: 42, d: 3.14}
...> |> Iteraptor.filter(fn {key, _} -> :c in key end, yield: :none)
%{a: %{e: %{c: 42}, d: %{c: 42}}, c: 42}
Flattening
iex> %{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}}
...> |> Iteraptor.to_flatmap(delimiter: "_")
#⇒ %{"a_b_c" => 42, "a_b_d_0" => nil, "a_b_d_1" => 42, "a_e_0" => :f, "a_e_1" => 42}

iex> %{"a.b.c": 42, "a.b.d.0": nil, "a.b.d.1": 42, "a.e.0": :f, "a.e.1": 42}
...> |> Iteraptor.from_flatmap
#⇒ %{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}}
Extras
iex> Iteraptor.Extras.bury([foo: :bar], ~w|a b c d|a, 42)
[a: [b: [c: [d: 42]]], foo: :bar]
In Details
Iterating
Iteraptor.each(term, fun/1, opts) — iterates the nested structure, yielding
the key and value. The returned from the function value is discarded.
	function argument: {key, value} tuple
	options: yield: [:all, :maps, :lists, :none], :none is the default
	return value: self

Mapping and Reducing
Iteraptor.map(term, fun/1, opts) — iterates the nested structure,
yielding the key and value. The value, returned from the block
should be either a single value or a {key, value} tuple.
	function argument: {key, value} tuple
	options: yield: [:all, :maps, :lists, :none], :none is the default
	return value: mapped

Iteraptor.reduce(term, fun/2, opts) — iterates the nested structure,
yielding the key and value. The value, returned from the block
should be an accumulator value.
	function arguments: {key, value}, acc pair
	options: yield: [:all, :maps, :lists, :none], :none is the default
	return value: accumulator

Iteraptor.map_reduce(term, fun/2, opts) — iterates the nested structure,
yielding the key and value. The value, returned from the block
should be a {​{key, value}, acc} value. The first element of this tuple is
used for mapping, the last—accumulating the result.
	function arguments: {key, value}, acc pair
	options: yield: [:all, :maps, :lists, :none], :none is the default
	return value: {mapped, accumulator} tuple

Filtering
Iteraptor.filter(term, filter/1, opts) — filters the structure
according to the value returned from each iteration (true to leave
the element, false to discard.)
	function argument: {key, value} tuple
	options: yield: [:all, :maps, :lists, :none], :none is the default
	return value: filtered

Flattening
Iteraptor.to_flatmap(term, opts) — flattens the structure into
the flatten map/keyword, concatenating keys with a delimiter.
	options: delimiter: binary(), into: term(),
defaults: delimiter: ".", into: %{}
	return value: flattened

Iteraptor.from_flatmap(term, fun/1, opts) — de-flattens the structure from
the flattened map/keyword, splitting keys by a delimiter. An optional transformer
function might be called after the value is deflattened.
	function argument: {key, value} tuple
	options: delimiter: binary(), into: term(),
defaults: delimiter: ".", into: %{}
	return value: Map.t | Keyword.t | List.t


  
    
Iteraptable protocol    



      
The protocol specifying how the respective struct might be used within Iteraptor.
Experimental. By implementing this protocol one might change the behaviour of
nested objects regarding how they should be iterated through.

    Returns a name of the term to be represented in flatmaps

      Link to this function
    
    name(term)

      
       
       View Source
     
  


  
      Specs

      
          name(term :: any()) :: binary()

      


Returns a name of the term to be represented in flatmaps
  


  
    
      
      Link to this function
    
    to_collectable(term)

      
       
       View Source
     
  


  
      Specs

      
          to_collectable(term :: any()) :: Collectable.t()

      


Converts a term to a collectable
  


  
    
      
      Link to this function
    
    to_enumerable(term)

      
       
       View Source
     
  


  
      Specs

      
          to_enumerable(term :: any()) :: Enumerable.t()

      


Converts a term to an enumerable
  


  
    
      
      Link to this function
    
    type(term)

      
       
       View Source
     
  


  
      Specs

      
          type(term :: any()) :: atom()

      


Returns a type understood by Iteraptable

Iteraptor makes complicated nested structures (currently Maps, Lists
  and Keywords) iteration easier.
Usage
Iterating, Mapping, Reducing
	Iteraptor.each/3
to iterate a deeply nested map/list/keyword;
	Iteraptor.map/3
to map a deeply nested map/list/keyword;
	Iteraptor.reduce/4
to reduce a deeply nested map/list/keyword;
	Iteraptor.map_reduce/4
to map and reduce a deeply nested map/list/keyword;

Flattening
	Iteraptor.to_flatmap/2
to flatten a deeply nested map/list/keyword into
flatten map with concatenated keys;
	Iteraptor.from_flatmap/3
to “unveil”/“unflatten” the previously flattened map into nested structure;

Filtering
	Iteraptor.filter/3
to filter the structure according to the value returned from each iteration
(true to leave the element, false to discard.)

    each(input, fun, opts \\ [])
  

    Iterates the given nested structure, calling the callback provided on each
  value. The key returned is an array of all the parent keys (and/or indices
  in a case of an array.)





  
    filter(input, fun, opts \\ [])
  

    Filters the deeply nested term, optionally calling the function on
filtered entries.





  
    from_flatmap(input, transformer \\ nil, opts \\ [])
  

    Build a nested structure out of a flatmap given, decomposing the names of keys
and handling lists carefully.





  
    jsonify(input, opts \\ [])
  

    Produces a term ready-to-use with JSON interchange. Stringifies all keys
and converts keywords to maps.





  
    map(input, fun, opts \\ [])
  

    Maps the given nested structure, calling the callback provided on each value.
  The key returned is a concatenated names of all the parent keys
(and/or indices in a case of an array.)





  
    map_reduce(input, acc \\ %{}, fun, opts \\ [])
  

    Iteration with mapping and reducing. The function of arity 2, called back on each
  iteration with {k, v} pair and an accumulator is accepted.





  
    reduce(input, acc \\ nil, fun, opts \\ [])
  

    Iteration with reducing. The function of arity 2, called back on each
  iteration with {k, v} pair and an accumulator is accepted.





  
    to_flatmap(input, opts \\ [])
  

    Build a flatmap out of nested structure, concatenating the names of keys.

      Link to this function
    
    each(input, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          each(
  %{} | Keyword.t() | [...] | Access.t(),
  ({any(), any()} -> any()),
  Keyword.t()
) :: %{} | Keyword.t() | [...] | Access.t()

      


Iterates the given nested structure, calling the callback provided on each
  value. The key returned is an array of all the parent keys (and/or indices
  in a case of an array.)
The return value is self.

  
  Parameters


	input: nested map/list/keyword to be walked through.
	fun: callback to be called on each {key, value} pair, where key
is an array or deeply nested keys;
e.g. on %{a: {b: 42}} will be called once, with tuple {[:a, :b], 42};
	opts: the options to be passed to the iteration
	yield: [:all | :maps | :keywords | nil] what to yield; default: nil
for yielding values only
	structs: [:values | :keep | nil] how to handle structs;  default: nil
for treating them as maps. When :values, the nested structs
are considered leaves and returned to the iterator instead of being iterated
through; when :keep it returns a struct back after iteration




  
  Examples


iex> %{a: %{b: %{c: 42}}} |> Iteraptor.each(&IO.inspect/1)
{[:a, :b, :c], 42}
%{a: %{b: %{c: 42}}}

iex> %{a: %{b: %{c: 42}}} |> Iteraptor.each(&IO.inspect/1, yield: :all)
{[:a], %{b: %{c: 42}}}
{[:a, :b], %{c: 42}}
{[:a, :b, :c], 42}
%{a: %{b: %{c: 42}}}
  


    
  
    
      
      Link to this function
    
    filter(input, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          filter(
  %{} | Keyword.t() | [...] | Access.t(),
  ({any(), any()} -> any()),
  Keyword.t()
) :: {%{} | Keyword.t() | [...] | Access.t(), any()}

      


Filters the deeply nested term, optionally calling the function on
filtered entries.
The return value is the filtered term.

  
  Parameters


	input: nested map/list/keyword to be filtered.
	fun: callback to be called on each {key, value} to filter entries.
	opts: the options to be passed to the iteration
	yield: [:all | :maps | :keywords | what to yield; default: nil
for yielding values only.




  
  Examples


iex> %{a: %{b: 42, e: %{f: 3.14, c: 42}, d: %{c: 42}}, c: 42, d: 3.14}
...> |> Iteraptor.filter(fn {key, _} -> :c in key end, yield: :none)
%{a: %{e: %{c: 42}, d: %{c: 42}}, c: 42}
  


    
    
  
    
      
      Link to this function
    
    from_flatmap(input, transformer \\ nil, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          from_flatmap(%{}, ({any(), any()} -> any()) | nil, Keyword.t()) ::
  %{} | [...] | Keyword.t()

      


Build a nested structure out of a flatmap given, decomposing the names of keys
and handling lists carefully.
%{"a.b.c": 42, "a.b.d.0": nil, "a.b.d.1": 42, "a.e.0": :f, "a.e.1": 42}
|> Iteraptor.from_flatmap
#⇒ %{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}}

  
  Parameters


	input: flat map to be “expanded” to nested maps/lists,  
	transformer: the transformer function to be called on all the elements,  
	opts: additional options to be passed through.


  
  Examples


iex> %{"a.b.c": 42} |> Iteraptor.from_flatmap
%{a: %{b: %{c: 42}}}

iex> %{"a.b.c": 42, "a.b.d": 42} |> Iteraptor.from_flatmap
%{a: %{b: %{c: 42, d: 42}}}

iex> %{"a.b.c": 42, "a.b.d": 42, "a.e": 42} |> Iteraptor.from_flatmap
%{a: %{b: %{c: 42, d: 42}, e: 42}}

iex> %{"0": 42, "1": 42} |> Iteraptor.from_flatmap
[42, 42]

iex> %{"1": :a1, "0": :a0, "2": :a2, "3": :a3, "4": :a4, "5": :a5,
...>   "6": :a6, "7": :a7, "8": :a8, "9": :a9, "10": :a10, "11": :a11}
...> |> Iteraptor.from_flatmap
[:a0, :a1, :a2, :a3, :a4, :a5, :a6, :a7, :a8, :a9, :a10, :a11]

iex> %{"0.a": 42, "0.b": 42} |> Iteraptor.from_flatmap
[%{a: 42, b: 42}]

iex> %{"a.0.0" => :b, "a.1" => 42, d: 42} |> Iteraptor.from_flatmap
%{a: [[:b], 42], d: 42}

iex> %{"a.b.c": 42, "a.b.d.0": nil, "a.b.d.1": 42, "a.e.0": :f, "a.e.1": 42}
...> |> Iteraptor.from_flatmap
%{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}}

iex> %{"0.a": 42, "0.b": 42} |> Iteraptor.from_flatmap(&IO.inspect/1)
{[0, :a], 42}
{[0, :b], 42}
[%{a: 42, b: 42}]
  


    
  
    
      
      Link to this function
    
    jsonify(input, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          jsonify(Access.container() | any(), opts :: list()) :: map()

      


Produces a term ready-to-use with JSON interchange. Stringifies all keys
and converts keywords to maps.
If the option keys: false is given, leaves keys intact.

  
  Examples


iex> Iteraptor.jsonify([foo: [:zzz], bar: :baz], values: true)
%{"foo" => ["zzz"], "bar" => "baz"}

iex> Iteraptor.jsonify(%{foo: [1, [bar: 2], 3], bar: [baz: 42]})
%{"foo" => [1, %{"bar" => 2}, 3], "bar" => %{"baz" => 42}}

iex> Iteraptor.jsonify([foo: [bar: [baz: :zoo], boo: 42]], values: true)
%{"foo" => %{"bar" => %{"baz" => "zoo"}, "boo" => 42}}

iex> Iteraptor.jsonify([foo: [bar: [baz: :zoo], boo: 42]], keys: false)
%{foo: %{bar: %{baz: :zoo}, boo: 42}}
  


    
  
    
      
      Link to this function
    
    map(input, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          map(
  %{} | Keyword.t() | [...] | Access.t(),
  ({any(), any()} -> any()),
  Keyword.t()
) :: %{} | Keyword.t() | [...] | Access.t()

      


Maps the given nested structure, calling the callback provided on each value.
  The key returned is a concatenated names of all the parent keys
(and/or indices in a case of an array.)
The return value is the result of subsequent calls to the transformer given.

  
  Parameters


	input: nested map/list/keyword to be mapped.
	fun: callback to be called on each {key, value} pair, where key
is an array or deeply nested keys;
e.g. on %{a: {b: 42}} will be called once, with tuple {[:a, :b], 42};
	opts: the options to be passed to the iteration
	yield: [:all | :maps | :keywords | what to yield; default: nil
for yielding values only.




  
  Examples


iex> %{a: %{b: %{c: 42}}} |> Iteraptor.map(fn {_, v} -> v * 2 end)
%{a: %{b: %{c: 84}}}

iex> %{a: %{b: %{c: 42}}} |> Iteraptor.map(fn {k, _} -> Enum.join(k) end)
%{a: %{b: %{c: "abc"}}}

iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.map(fn
...>      {[_], _} = self -> self
...>      {[_, _], _} -> "YAY"
...>    end, yield: :all)
%{a: %{b: "YAY"}}
  


    
    
  
    
      
      Link to this function
    
    map_reduce(input, acc \\ %{}, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          map_reduce(
  %{} | Keyword.t() | [...] | Access.t(),
  %{} | Keyword.t() | [...] | Access.t(),
  ({any(), any()}, any() -> any()),
  Keyword.t()
) :: {%{} | Keyword.t() | [...] | Access.t(), any()}

      


Iteration with mapping and reducing. The function of arity 2, called back on each
  iteration with {k, v} pair and an accumulator is accepted.
The return value is the tuple, consisting of mapped input and the
  accumulator from the last call to the passed map-reducer.

  
  Parameters


	input: nested map/list/keyword to be mapped.
	fun: callback to be called on each {key, value}, acc pair,
where key is an array or deeply nested keys, value is the value and
acc is the accumulator;
	opts: the options to be passed to the iteration
	yield: [:all | :maps | :keywords | what to yield; default: nil
for yielding values only.




  
  Examples


iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.map_reduce([], fn
...>      {k, %{} = v}, acc -> {{k, v}, [Enum.join(k, ".") | acc]}
...>      {k, v}, acc -> {{k, v * 2}, [Enum.join(k, ".") <> "=" | acc]}
...>    end, yield: :all)
{%{a: %{b: %{c: 84}}}, ["a.b.c=", "a.b", "a"]}
  


    
    
  
    
      
      Link to this function
    
    reduce(input, acc \\ nil, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          reduce(
  %{} | Keyword.t() | [...] | Access.t(),
  %{} | Keyword.t() | [...] | Access.t(),
  ({any(), any()}, any() -> any()),
  Keyword.t()
) :: {%{} | Keyword.t() | [...] | Access.t(), any()}

      


Iteration with reducing. The function of arity 2, called back on each
  iteration with {k, v} pair and an accumulator is accepted.
The return value is the result of the last call to the passed reducer function.

  
  Parameters


	input: nested map/list/keyword to be mapped.
	fun: callback to be called on each {key, value}, acc pair,
where key is an array or deeply nested keys, value is the value and
acc is the accumulator;
	opts: the options to be passed to the iteration
	yield: [:all | :maps | :keywords | what to yield; default: nil
for yielding values only.




  
  Examples


iex> %{a: %{b: %{c: 42}}}
...> |> Iteraptor.reduce([], fn {k, _}, acc ->
...>      [Enum.join(k, "_") | acc]
...>    end, yield: :all)
...> |> :lists.reverse()
["a", "a_b", "a_b_c"]
  


    
  
    
      
      Link to this function
    
    to_flatmap(input, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          to_flatmap(%{} | [...], Keyword.t()) :: %{}

      


Build a flatmap out of nested structure, concatenating the names of keys.
%{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}} |> Iteraptor.to_flatmap
%{"a.b.c": 42, "a.b.d.0": nil, "a.b.d.1": 42, "a.e.0": :f, "a.e.1": 42}
Lists are handled gracefully, index is used as a key in resulting map.

  
  Parameters


	input: nested map/list/keyword/struct to be flattened,  
	opts: the additional options to be passed through:  
  — delimiter (default: ".",) might be passed explicitly or
configured with :iteraptor, :delimiter application setting.


  
  Examples


iex> [:a, 42] |> Iteraptor.to_flatmap
%{0 => :a, 1 => 42}

iex> %{a: 42} |> Iteraptor.to_flatmap
%{a: 42}

iex> %{a: 42, b: 42} |> Iteraptor.to_flatmap
%{a: 42, b: 42}

iex> %{a: %{b: 42}, d: 42} |> Iteraptor.to_flatmap
%{"a.b" => 42, d: 42}

iex> %{a: [:b, 42], d: 42} |> Iteraptor.to_flatmap
%{"a.0" => :b, "a.1" => 42, d: 42}

iex> %{a: %{b: [:c, 42]}, d: 42} |> Iteraptor.to_flatmap
%{"a.b.0" => :c, "a.b.1" => 42, d: 42}

iex> %{a: %{b: 42}} |> Iteraptor.to_flatmap
%{"a.b" => 42}

iex> %{a: %{b: %{c: 42}}} |> Iteraptor.to_flatmap
%{"a.b.c" => 42}

iex> %{a: %{b: %{c: 42}}, d: 42} |> Iteraptor.to_flatmap
%{"a.b.c" => 42, d: 42}

iex> [a: [b: [c: 42]], d: 42] |> Iteraptor.to_flatmap
%{"a.b.c" => 42, d: 42}

iex> [a: [[:b], 42], d: 42] |> Iteraptor.to_flatmap
%{"a.0.0" => :b, "a.1" => 42, d: 42}

iex> %{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}} |> Iteraptor.to_flatmap
%{"a.b.c" => 42, "a.b.d.0" => nil, "a.b.d.1" => 42, "a.e.0" => :f, "a.e.1" => 42}

iex> %{a: %{b: %{c: 42, d: [nil, 42]}, e: [:f, 42]}}
...> |> Iteraptor.to_flatmap(delimiter: "_")
%{"a_b_c" => 42, "a_b_d_0" => nil, "a_b_d_1" => 42, "a_e_0" => :f, "a_e_1" => 42}
  

        

      
  
    
Iteraptor.Array    



      
Array emulation implementing Access behaviour. Index in array is zero-based.
Array is the "go to" array data structure in Elixir. An array can be
  constructed using Array.new/{0,1}:
iex> Iteraptor.Array.new()
#Array<[]>

iex> Iteraptor.Array.new(2)
#Array<[nil, nil]>

iex> Iteraptor.Array.new([:foo, :bar])
#Array<[:foo, :bar]>
An array can contain any kind of elements, and elements in an array don't have
  to be of the same type. By definition, arrays have keys in 0..size-1 range.
Arrays are implicitly expandable, which means adding an element at index 100
  to the array currently containing 1 element would increase the size of the
  array to 100.
iex> array = Iteraptor.Array.new([:foo])
iex> Iteraptor.Array.set(array, 3, :bar)
#Array<[:foo, nil, nil, :bar]>
An Array is represented internally using the %Array{} struct. Note that,
  however, the struct fields are private and must not be accessed directly;
  use the functions in this module to perform operations on arrays.
Arrays can also be constructed starting from other collection-type data
structures: for example, see Array.new/1 or Enum.into/2.
iex> Enum.into([1, 2, 3], Iteraptor.Array.new())
#Array<[1, 2, 3]>
Arrays do implement Access behaviour.
iex> array = Iteraptor.Array.new([%{foo: 42}, %{bar: :baz}])
iex> get_in(array, [0, :foo])
42
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Types
    


  
    t()
  




  
    value()
  



  

  
    
      Functions
    


  
    append(array, other)
  

    Appends another enumerable to the array.





  
    from_tuple(tuple)
  

    Converts a tuple given as parameter to array.





  
    get(array, index, default \\ nil)
  

    Returns the value at index in array, or default if index is out of array bounds.





  
    new(enumerable \\ nil, transform \\ nil)
  

    Creates an array of the given length or from enumerable. Might we used to wrap
the existing instance of Iteraptor.Array.





  
    pop(array, index)
  

    Pops (deletes) value at index from array, setting the value at the
  respective index to nil.
Returns a tuple containing the value removed and the new array.





  
    set(array, index, value)
  

    Sets the value at index in array, expanding the array if necessary.
Returns a new array.





  
    size(array)
  

    Returns the number of elements in array.





  
    to_list(array)
  

    Converts array to a list.





  
    trim(array)
  

    Trims nil values from the tail of the Array. Returns a trimmed array.




  

      

      
        
          
            
            Anchor for this section
          
Types        

        

  
    
      
      Link to this type
    
    t()

      
       
       View Source
     
  


  
      Specs

      
          t() :: %Iteraptor.Array{
  map: %{required(non_neg_integer()) => any()},
  version: term()
}

      


  


  
    
      
      Link to this type
    
    value()

      
       
       View Source
     
  


  
      Specs

      
          value() :: term()

      


  

        

      
      
        
          
            
            Anchor for this section
          
Functions        

        

  
    
      
      Link to this function
    
    append(array, other)

      
       
       View Source
     
  


  
      Specs

      
          append(t(), any()) :: t()

      


Appends another enumerable to the array.
iex> array = Iteraptor.Array.new([1, 2, 3])
iex> Iteraptor.Array.append(array, [4, 5])
#Array<[1, 2, 3, 4, 5]>
  


  
    
      
      Link to this function
    
    from_tuple(tuple)

      
       
       View Source
     
  


  
      Specs

      
          from_tuple(tuple :: tuple()) :: t()

      


Converts a tuple given as parameter to array.
iex> Iteraptor.Array.from_tuple({1, 2, 3})
#Array<[1, 2, 3]>
  


    
  
    
      
      Link to this function
    
    get(array, index, default \\ nil)

      
       
       View Source
     
  


  
      Specs

      
          get(t(), non_neg_integer(), any()) :: any()

      


Returns the value at index in array, or default if index is out of array bounds.
iex> array = Iteraptor.Array.new([42])
iex> Iteraptor.Array.get(array, 0)
42
iex> Iteraptor.Array.get(array, 2, 42)
42
  


    
    
  
    
      
      Link to this function
    
    new(enumerable \\ nil, transform \\ nil)

      
       
       View Source
     
  


  

Creates an array of the given length or from enumerable. Might we used to wrap
the existing instance of Iteraptor.Array.
iex> Iteraptor.Array.new(3)
#Array<[nil, nil, nil]>

iex> Iteraptor.Array.new([:foo, :bar, 42])
#Array<[:foo, :bar, 42]>
Also the transformation function might be passed via second argument.
iex> Iteraptor.Array.new([1, 2, 3], fn x -> 2 * x end)
#Array<[2, 4, 6]>
  


  
    
      
      Link to this function
    
    pop(array, index)

      
       
       View Source
     
  


  
      Specs

      
          pop(t(), non_neg_integer()) :: {any(), t()}

      


Pops (deletes) value at index from array, setting the value at the
  respective index to nil.
Returns a tuple containing the value removed and the new array.
iex> array = Iteraptor.Array.new([1, 2, 3])
iex> {elem, array} = Iteraptor.Array.pop(array, 1)
iex> elem
2
iex> array
#Array<[1, nil, 3]>
  


  
    
      
      Link to this function
    
    set(array, index, value)

      
       
       View Source
     
  


  
      Specs

      
          set(t(), non_neg_integer(), any()) :: t()

      


Sets the value at index in array, expanding the array if necessary.
Returns a new array.
iex> array = Iteraptor.Array.new([42])
iex> Iteraptor.Array.set(array, 0, :foo)
#Array<[:foo]>
iex> Iteraptor.Array.set(array, 2, :bar)
#Array<[42, nil, :bar]>
  


  
    
      
      Link to this function
    
    size(array)

      
       
       View Source
     
  


  
      Specs

      
          size(t()) :: non_neg_integer()

      


Returns the number of elements in array.
iex> Iteraptor.Array.size(Iteraptor.Array.new([1, 2, 3]))
3
  


  
    
      
      Link to this function
    
    to_list(array)

      
       
       View Source
     
  


  
      Specs

      
          to_list(t()) :: [any()]

      


Converts array to a list.
iex> Iteraptor.Array.to_list(Iteraptor.Array.new([1, 2, 3]))
[1, 2, 3]
  


  
    
      
      Link to this function
    
    trim(array)

      
       
       View Source
     
  


  
      Specs

      
          trim(array :: t()) :: t()

      


Trims nil values from the tail of the Array. Returns a trimmed array.
iex> array = Iteraptor.Array.new([42, nil, nil])
#Array<[42, nil, nil]>
iex> Iteraptor.Array.trim(array)
#Array<[42]>
  

        

      
  
    
Iteraptor.Config    



      
  Extra functions to deal with runtime configuration.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    from_env(app, converter \\ &(&1), key \\ :system)
  

    Reads the application config and recursively changes all the {:key, "VAR"}
  tuples with runtime system environment read from "VAR".




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
    
  
    
      
      Link to this function
    
    from_env(app, converter \\ &(&1), key \\ :system)

      
       
       View Source
     
  


  
      Specs

      
          from_env(app :: atom(), converter :: (binary() -> any()), key :: atom()) :: :ok

      


Reads the application config and recursively changes all the {:key, "VAR"}
  tuples with runtime system environment read from "VAR".
Default value for :key is :system, but it might be adjusted to avoid
  clashes with other libraries implementing {:system, "FOO"} functionality,
  like Phoenix and/or Ecto.

  
  Examples:


iex> System.put_env("FOO", "42")
iex> Application.put_env(
...>   :iteraptor,
...>   :key,
...>   [value: [value: {:my_system, "FOO"}]],
...>   persistent: true
...> )
iex> Application.get_all_env(:iteraptor)
[key: [value: [value: {:my_system, "FOO"}]]]
iex> Iteraptor.Config.from_env(:iteraptor, &String.to_integer/1, :my_system)
iex> Application.get_all_env(:iteraptor)
[key: [value: [value: 42]]]
  

        

      
  
    
Iteraptor.Extras    



      
  Extra functions to deal with enumerables.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    bury(term, key, value, opts \\ [into: :default])
  

    Deep store the value into the nested structure. Behaves as a proposed
  but rejected in ruby core
  Hash#bury.





  
    each_cons(list, n \\ 2, acc \\ [])
  

    Behaves as Enum.each_cons(n) in ruby. Iterates the input producing the list of cons.
  Gracefully stolen from https://groups.google.com/forum/#!topic/elixir-lang-core/LAK23vaJgvE




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this function
    
    bury(term, key, value, opts \\ [into: :default])

      
       
       View Source
     
  


  
      Specs

      
          bury(
  Access.t(),
  [...],
  any(),
  [{:into, :default | :map | :keyword}] | Keyword.t()
) :: Access.t()

      


Deep store the value into the nested structure. Behaves as a proposed
  but rejected in ruby core
  Hash#bury.

  
  Examples:


iex> Iteraptor.Extras.bury(%{foo: :bar}, ~w|a b c d|a, 42)
%{a: %{b: %{c: %{d: 42}}}, foo: :bar}
iex> Iteraptor.Extras.bury([foo: :bar], ~w|a b c d|a, 42)
[a: [b: [c: [d: 42]]], foo: :bar]
iex> Iteraptor.Extras.bury(%{foo: :bar}, ~w|a b c d|a, 42, into: :keyword)
%{a: [b: [c: [d: 42]]], foo: :bar}
iex> Iteraptor.Extras.bury(42, ~w|a b c d|a, 42)
** (Iteraptor.Utils.Unsupported) Unsupported term 42 in call to Iteraptor.Extras.bury/4.
  


    
    
  
    
      
      Link to this function
    
    each_cons(list, n \\ 2, acc \\ [])

      
       
       View Source
     
  


  
      Specs

      
          each_cons([...] | %{} | binary(), integer(), Keyword.t()) :: [...]

      


Behaves as Enum.each_cons(n) in ruby. Iterates the input producing the list of cons.
  Gracefully stolen from https://groups.google.com/forum/#!topic/elixir-lang-core/LAK23vaJgvE

  
  Examples


iex> 'letters' |> Iteraptor.Extras.each_cons
['le', 'et', 'tt', 'te', 'er', 'rs']
iex> 'letters' |> Iteraptor.Extras.each_cons(4)
['lett', 'ette', 'tter', 'ters']
iex> 1..6 |> Iteraptor.Extras.each_cons(4)
[[1,2,3,4], [2,3,4,5], [3,4,5,6]]
iex> "letters" |> Iteraptor.Extras.each_cons(3)
["let", "ett", "tte", "ter", "ers"]
  

        

      
  
    
Iteraptor.Iteraptable    



      
use Iteraptor.Iteraptable inside structs to make them both
Enumerable and
Collectable and
implement the Access behaviour:
Usage
Use the module within the struct of your choice and this struct will be
automagically granted Enumerable and Collectable protocols implementations.
use Iteraptor.Iteraptable accepts keyword parameter skip: Access or
skip: [Enumerable, Collectable] which allows to implement a subset of
protocols. Also it accepts keyword parameter derive: MyProtocol allowing
to specify what protocol(s) implementations should be implicitly derived
for this struct.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    __using__(opts \\ [])
  

    Allows to enable iterating features on structs with use Iteraptor.Iteraptable




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this macro
    
    __using__(opts \\ [])

      
       
       View Source
     
      (macro)
  


  

Allows to enable iterating features on structs with use Iteraptor.Iteraptable

  
  Parameters


	keyword parameter opts
	skip: Access or skip: [Enumerable, Collectable] allows
to implement a subset of protocols;
	derive: MyProtocol allows to derive selected protocol implementation(s).



  

        

      
  
    
Iteraptor.AST    



      
Iteraptor.AST module traverses AST, allowing map, reduce and family.
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    map(input, fun, opts \\ [])
  

    Mapper for the AST.





  
    reduce(input, acc, fun, opts \\ [])
  

    Reduces the AST with an accumulator.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this function
    
    map(input, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          map(
  binary() | {atom(), list(), any()} | list(),
  (any(), any() -> any()),
  list()
) :: any()

      


Mapper for the AST.

  
  Parameters


	input: the AST to traverse
	fun: the function to be called on the tree element
	opts: the options to be passed to the iteration
	yield: :all | nil what to yield; default: nil
for yielding values only.




  
  Examples


iex> bindings = [a: 1, b: 2, c: 3]
...> ":math.sin(42 * a / (3.14 * b)) > c"
...> |> Iteraptor.AST.map(fn
...>      {var, _, val} when is_atom(val) -> bindings[var]
...>      any -> any
...>    end)
{:>, [line: 1],
  [
    {{:., [line: 1], [:math, :sin]}, [line: 1],
      [
        {:/, [line: 1],
        [
          {:*, [line: 1], [42, 1]},
          {:*, [line: 1], [3.14, 2]}
        ]}
      ]},
    3
  ]}
  


    
  
    
      
      Link to this function
    
    reduce(input, acc, fun, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          reduce(
  binary() | {atom(), list(), any()} | list(),
  any(),
  (any(), any() -> any()),
  list()
) :: any()

      


Reduces the AST with an accumulator.

  
  Parameters


	input: the AST to traverse
	acc: the accumulator
	fun: the function to be called on the tree element
	opts: the options to be passed to the iteration
	yield: :all | nil what to yield; default: nil
for yielding values only.




  
  Examples


iex> ":math.sin(42 * a / (3.14 * b)) > c"
...> |> Iteraptor.AST.reduce([], fn
...>      {var, _, val}, acc when is_atom(val) -> [var | acc]
...>      _, acc -> acc
...>    end)
...> |> Enum.reverse()
~w|a b c|a
  

        

      
  
    
Iteraptor.Utils    



      
Helper functions to update nested terms
      

      
        
          
            
            Anchor for this section
          
          Summary
        

  
    
      Functions
    


  
    deep_put_in(target, key_value, opts \\ [])
  

    Safe put the value deeply into the term nesting structure. Creates
all the intermediate keys if needed.





  
    dig(input, acc \\ [])
  

    Digs the leaf value in the nested keyword / map.





  
    dig!(input, acc \\ [])
  




  
    join(input, opts \\ [])
  

    Joins the array of keys into the string using delimiter.





  
    quacks_as_list(input)
  

    Checks if the map/keyword looks like a normal list.





  
    split(input, opts \\ [])
  

    Splits the string by delimiter, possibly converting the keys to symbols.





  
    squeeze(input, opts \\ [])
  

    Squeezes the nested structure merging same keys.





  
    try_to_list(input)
  

    Gently tries to create a linked list out of input, returns input if it
  cannot be safely converted to the list.





  
    type(input)
  

    Determines the type of the given term.




  

      

      
        
          
            
            Anchor for this section
          
Functions        

        

    
  
    
      
      Link to this function
    
    deep_put_in(target, key_value, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          deep_put_in(%{} | keyword(), {list(), any()}, keyword()) :: %{} | keyword()

      


Safe put the value deeply into the term nesting structure. Creates
all the intermediate keys if needed.

  
  Examples:


iex> Iteraptor.Utils.deep_put_in(%{}, {~w|a b c|a, 42})
%{a: %{b: %{c: 42}}}
iex> Iteraptor.Utils.deep_put_in(%{a: %{b: %{c: 42}}}, {~w|a b d|a, :foo})
%{a: %{b: %{c: 42, d: :foo}}}
iex> Iteraptor.Utils.deep_put_in(%{a: %{b: [c: 42]}}, {~w|a b d|a, :foo})
%{a: %{b: [c: 42, d: :foo]}}
iex> Iteraptor.Utils.deep_put_in(%{a: %{b: [42]}}, {~w|a b|a, :foo})
%{a: %{b: [42, :foo]}}
iex> Iteraptor.Utils.deep_put_in(%{a: [:foo, %{b: 42}]}, {~w|a b|a, :foo})
%{a: [:foo, %{b: 42}, {:b, :foo}]}
  


    
  
    
      
      Link to this function
    
    dig(input, acc \\ [])

      
       
       View Source
     
  


  
      Specs

      
          dig(%{} | keyword(), keyword()) :: {:ok, {list(), any()}} | {:error, any()}

      


Digs the leaf value in the nested keyword / map.

  
  Examples:


iex> Iteraptor.Utils.dig(%{k1: %{k2: %{k3: :value}}})
{:ok, {[:k1, :k2, :k3], :value}}
iex> Iteraptor.Utils.dig([k1: [k2: [k3: :value]]])
{:ok, {[:k1, :k2, :k3], :value}}
iex> Iteraptor.Utils.dig([k1: :value, k2: :value])
{:error, [k1: :value, k2: :value]}
iex> Iteraptor.Utils.dig([k1: %{k2: [k3: :value]}])
{:ok, {[:k1, :k2, :k3], :value}}
  


    
  
    
      
      Link to this function
    
    dig!(input, acc \\ [])

      
       
       View Source
     
  


  
      Specs

      
          dig!(%{} | keyword(), keyword()) :: {list(), any()} | no_return()

      


  


    
  
    
      
      Link to this function
    
    join(input, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          join(Enum.t(), keyword()) :: binary()

      


Joins the array of keys into the string using delimiter.

  
  Examples:


iex> Iteraptor.Utils.join(~w|a b c d|)
"a.b.c.d"
iex> Iteraptor.Utils.join(~w|a b c d|, delimiter: "_")
"a_b_c_d"
  


  
    
      
      Link to this function
    
    quacks_as_list(input)

      
       
       View Source
     
  


  
      Specs

      
          quacks_as_list(%{} | keyword() | any()) :: true | false

      


Checks if the map/keyword looks like a normal list.

  
  Examples:


iex> Iteraptor.Utils.quacks_as_list(%{"0" => :foo, 1 => :bar})
true
iex> Iteraptor.Utils.quacks_as_list([{:"1", :bar}, {:"0", :foo}])
true
iex> Iteraptor.Utils.quacks_as_list(%{foo: :bar})
false
iex> Iteraptor.Utils.quacks_as_list(%{"5" => :foo, "1" => :bar})
false
iex> Iteraptor.Utils.quacks_as_list(42)
false
  


    
  
    
      
      Link to this function
    
    split(input, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          split(input :: binary(), opts :: keyword()) :: [binary() | atom()]

      


Splits the string by delimiter, possibly converting the keys to symbols.

  
  Examples:


iex> Iteraptor.Utils.split("a.b.c.d", transform: :none)
["a", "b", "c", "d"]
iex> Iteraptor.Utils.split("a_b_c_d", delimiter: "_")
["a", "b", "c", "d"]
iex> Iteraptor.Utils.split("a.b.c.d", transform: :unsafe)
[:a, :b, :c, :d]
iex> Iteraptor.Utils.split("a.b.c.d", transform: :safe)
[:a, :b, :c, :d]
  


    
  
    
      
      Link to this function
    
    squeeze(input, opts \\ [])

      
       
       View Source
     
  


  
      Specs

      
          squeeze(%{} | keyword() | list() | Access.t(), keyword()) ::
  %{} | keyword() | list()

      


Squeezes the nested structure merging same keys.

  
  Examples:


#iex> Iteraptor.Utils.squeeze([foo: [bar: 42], foo: [baz: 3.14]])
#[foo: [bar: 42, baz: 3.14]]
iex> Iteraptor.Utils.squeeze([foo: %{bar: 42}, foo: %{baz: 3.14}])
[foo: %{bar: 42, baz: 3.14}]
iex> Iteraptor.Utils.squeeze([foo: %{bar: 42}, foo: :baz])
[foo: [%{bar: 42}, :baz]]
iex> Iteraptor.Utils.squeeze([a: [b: [c: 42]], a: [b: [d: 3.14]]])
[a: [b: [c: 42, d: 3.14]]]
iex> Iteraptor.Utils.squeeze([a: [b: [c: 42]], a: [b: %{d: 3.14}]])
[a: [b: [c: 42, d: 3.14]]]
iex> Iteraptor.Utils.squeeze([a: [b: [c: :foo]], a: [b: [c: 3.14]]])
[a: [b: [c: [:foo, 3.14]]]]
iex> Iteraptor.Utils.squeeze([a: [b: [:foo, :bar]], a: [b: [c: 3.14]]])
[a: [b: [:foo, :bar, {:c, 3.14}]]]
iex> Iteraptor.Utils.squeeze([a: [:foo, :bar], a: [b: [c: 3.14]]])
[a: [:foo, :bar, {:b, [c: 3.14]}]]
  


  
    
      
      Link to this function
    
    try_to_list(input)

      
       
       View Source
     
  


  
      Specs

      
          try_to_list(any()) :: list() | any()

      


Gently tries to create a linked list out of input, returns input if it
  cannot be safely converted to the list.

  
  Examples:


iex> Iteraptor.Utils.try_to_list(%{"0" => :foo, 1 => :bar})
[:foo, :bar]
iex> Iteraptor.Utils.try_to_list([{:"1", :bar}, {:"0", :foo}])
[:foo, :bar]
iex> Iteraptor.Utils.try_to_list(%{foo: :bar})
%{foo: :bar}
iex> Iteraptor.Utils.try_to_list(%{"5" => :foo, "1" => :bar})
%{"5" => :foo, "1" => :bar}
  


  
    
      
      Link to this function
    
    type(input)

      
       
       View Source
     
  


  
      Specs

      
          type(%{} | keyword() | list() | any()) :: {atom(), any(), any()} | :error

      


Determines the type of the given term.

  
  Examples:


iex> Iteraptor.Utils.type(%{foo: :bar})
{Map, %{foo: :bar}, %{}}
iex> Iteraptor.Utils.type([foo: :bar])
{Keyword, [foo: :bar], []}
iex> Iteraptor.Utils.type([{:foo, :bar}])
{Keyword, [{:foo, :bar}], []}
iex> Iteraptor.Utils.type(~w|foo bar|a)
{List, [:foo, :bar], []}
iex> Iteraptor.Utils.type(42)
:error
  

        

      
  
    
Iteraptor.Iteraptable.Unsupported exception    



      
Unsupported in applying Iteraptor.Iteraptable
      


  
    
Iteraptor.Utils.Unsupported exception    



      
An exception to be thrown from banged methods of Iteraptor.
Sooner or later we’ll support everything, that’s why meanwhile
  we raise Unsupported if something goes wrong.
      


  
    
Iteraptor.Iteraptable.Unsupported exception    



      
Unsupported in applying Iteraptor.Iteraptable
      


  
    
Iteraptor.Utils.Unsupported exception    



      
An exception to be thrown from banged methods of Iteraptor.
Sooner or later we’ll support everything, that’s why meanwhile
  we raise Unsupported if something goes wrong.

